

# De muis

Cursus ML

MSX Computer & Club Magazine nummer 64 - december '93 /januari '94

*Scanned, ocr'ed and converted to PDF by HansO, 2001*

Een beetje menugestuurd programma moet tegenwoordig met een muis kunnen werken. In deze cursus zullen we eens kijken hoe we de muis vanuit machinetaal kunnen aansturen. Natuurlijk inclusief een kant en klare listing voor muisgebruik.

We willen graag een kleine wending maken met deze cursus. In plaats van algemene theorie, zullen we nu speciale onderwerpen behandelen en daarbij een complete listing geven, die je zelf kunt gebruiken maar in ieder geval duidelijk laat zien hoe je bepaalde functies en/of mogelijkheden kunt aanspreken. De reden hiervoor is, dat we in de loop van de tijd—en dat gaat ver terug tot in de MSX Club historie—al zo'n beetje alles behandeld hebben, van geluid tot video en van BIOS en Math-Pack tot de hooks.

## Muis in BIOS

Deze keer zullen we een blik werpen op een routine die de muis kan aansturen. In de BIOS bevindt zich een routine, te weten GTPAD (&HDB), om de coördinaten van de muis uit te lezen. Maar het is niet zo simpel, dat je deze coördinaten rechtstreeks kunt opvragen. De routine heeft maar liefst negentien functies. Voor de muis zijn functies 12 tot en met 19 van belang. De andere hebben betrekking op touch panel en lichtpen.

De functies 12 tot en met 15 zijn voor de muis in port 1 en de functies van 16 tot en met 19 zijn dezelfde maar dan voor port 2. Bij het aanroepen van de routine GTPAD staat het functienummer in register A en de terugkerende waarde staat ook in A. Functie 12 geeft altijd &HFF terug, maar is bedoeld als initialisatie-functie. Voor je de coördinaten uitleest, moet je eerst deze functie aanroepen, anders zijn de waarden van de coördinaten niet geldig. De periode tussen het aanroepen van functie 12 en het eigenlijke uitlezen van de coördinaten moet zo kort mogelijk zijn om de waarden betrouwbaar te houden.

De functies 13 en 14 geven dan respectievelijk de X- en de Y-coördinaat. Deze coördinaten zijn echter relatief, dat wil zeggen dat ze alleen het afgelegde pad van de muis vertegenwoordigen. Om ze tot absolute coördinaten om te rekenen moet je ze dus optellen bij de coördinaten van de vorige keer.

## Soepel

Het is van belang voor een soepele muisbediening dat de coördinaten regelmatig uitgelezen worden. Aangezien de routine niet erg veel tijd kost, is het aan te raden deze op interrupt te zetten en tevens de muiscursor op interrupt af te beelden.

## Binnen kader

De grootste problemen doen zich voor bij het binnen een kader houden van de muiscursor. In de listing zie je een methode om dit effect te realiseren. Er wordt voor zowel de X- als de Y-coördinaat gekeken welke kant de muis bewogen wordt en van daar uit wordt gekeken of hij de grens overschrijdt. Is dit het geval dan krijgt de coördinaat de waarde van grens minus (of plus) één pixel. Hij mag namelijk de grenswaarde niet passeren.

Je kunt de begincoördinaten vaststellen door deze aan het begin in CUR\_X en CUR\_Y te zetten. Het kader kun je definiëren door X\_MIN, X\_MAX, Y\_MIN en Y\_MAX de juiste waarden te geven. Natuurlijk kun je die tijdens de uitvoering van je programma veranderen. Daarom hebben we ze ook niet als EQUates gedefinieerd.

## Klikken

Om te kijken of er geklikt wordt, kun je dezelfde routine als voor de vuurknop van de joystick en het toetsenbord (de spatiebalk) gebruiken, namelijk GTTRG (&HD8). In A geef je het poortnummer aan (1 of 2 voor joystick of muis, of 0 voor spatiebalk). Is de uitkomst (uiteraard ook in A) nul dan is de knop niet ingedrukt en is de uitkomst 255 dan is hij wel ingedrukt.

```
; MOUSE.SRC
; Muis-driver voor eigen gebruik
; Door: Falco Dam en Ivo Wubbels
; (c) 1993 MSX-Engine
DEFB OFEH
DEFW START
DEFW M_PORT
DEFW START
ORG      0C000H
START:   JP RDMOUS      ; coördinaten muiscursor (&hC003 en &hC004)
CUR_X:   DEFB 128
CUR_Y:   DEFB 96
GTPAD    EQU 000BH
RQ_VAL   EQU 12
MOUSEX   EQU 13
MOUSEY   EQU 14
RDMOUS:  LD A,RQ_VAL
          LD HL,M_PORT      ; JUISTE POORT
          ADD A,(HL)
          PUSH HL
          CALL GTPAD        ; VRAAG COÖRDINATEN AAN
          POP HL
          LD A,MOUSEX
          ADD A,(HL)        ; JUISTE POORT
          CALL GTPAD        ; LEES RELATIEVE X
          LD HL,CUR_X
          AND A              ; WELKE RICHTING?
          JP M,X_NEG        ; LINKS12
          ADD A,(HL)        ; RECHTS
          JR C,S_XMAX       ; OVER SCHERMBORDER
          LD HL,X__MAX
          CP (HL)           ; BUITEN KADER?
          JP C,ST_X
S_XMAX:  LD A,(X_MAX)       ; NEEM GRENSWAARDE
ST_X:    LD (CUR_X),A      ; SLA X OP
          LD A,MOUSEY
          LD HL,M_PORT      ; JUISTE POORT
          ADD A,(HL)
          CALL GTPAD        ; LEES RELATIEVE Y
          LD HL,CUR_Y
          AND A              ; WELKE RICHTING?
          JP M,Y_NEG        ; BOVEN
          ADD A,(HL)        ; ONDER
          JP C,S_YMAX       ; OVER SCHERMBORDER
          LD HL,Y_MAX
          CP (HL)           ; BUITEN KADER?
          JP C,ST_Y
S_YMAX:  LD A,(HL)         ; NEEM GRENSWAARDE
ST_Y:    LD (CUR_Y),A      ; SLA Y OP
          RET
```

```

X_NEG:      ADD A,(HL)
            JP NC,S_XMIN          ; OVER SCHERMBORDER
            LD HL,X_MIN
            CP (HL)              ; BUITEN KADER?
            JP NC,ST_X
S_XMIN:     LD A,(X_MIN)          ; NEEM GRENSWAARDE+1
            INC A
            JP ST_X              ; SLA X OP
Y_NEG:      ADD A,(HL)
            JP NC,S_YMIN         ; OVER SCHERMBORDER
            LD HL,Y_MIN
            CP (HL)              ; BUITEN KADER?
            JP NC,ST_Y
S_YMIN:     LD A,(Y_MIN)          ; NEEM GRENSWAARDE+1
            INC A
            JP ST_Y              ; SLA Y OP
; Kader waarbinnen muis moet blijven
X_MIN:     DEFB 0
X_MAX:     DEFB 255
Y_MIN:     DEFB 0
Y_MAX:     DEFB 211
; muispoort (0=port 1, 4=port 2)
M_PORT:    DEFB 0

```